



Project Showcase: **On-chain RPG** (Part 1)

March 13, 2023

Agenda

- 01 **Objects**
- 02 **Events**
- 03 **Initialization**
- 04 **Gameplay**

- 05 **Inventory**
- 06 **Object Creation**
- 07 **Game Integrity / Links Checks**

Game Time!



In a new terminal enter: `sui move new onchain_rpg`

Adapted from:

github.com/MystenLabs/sui/blob/main/sui_programmability/examples/games/sources/hero.move



```
1  module onchain_rpg::hero {  
2      use sui::coin::{Self, Coin};  
3      use sui::event;  
4      use sui::object::{Self, ID, UID};  
5      use sui::math;  
6      use sui::sui::SUI;  
7      use sui::transfer;  
8      use sui::tx_context::{Self, TxContext};  
9      use std::option::{Self, Option};
```

Objects



```
1  /// Our hero!
2  struct Hero has key, store {
3      id: UID,
4      /// Hit points. If they go to zero, the hero can't do anything
5      hp: u64,
6      /// Experience of the hero. Begins at zero
7      experience: u64,
8      /// The hero's minimal inventory
9      sword: Option<Sword>,
10     /// An ID of the game user is playing
11     game_id: ID,
12 }
```



```
1  /// The hero's trusty sword
2  struct Sword has key, store {
3      id: UID,
4      /// Constant set at creation. Acts as a multiplier on sword's strength.
5      /// Swords with high magic are rarer (because they cost more).
6      magic: u64,
7      /// Sword grows in strength as we use it
8      strength: u64,
9      /// An ID of the game
10     game_id: ID,
11 }
```



```
1  /// For healing wounded heroes
2  struct Potion has key, store {
3      id: UID,
4      /// Effectiveness of the potion
5      potency: u64,
6      /// An ID of the game
7      game_id: ID,
8  }
```




```
1  /// A creature that the hero can slay to level up
2  struct Boar has key {
3      id: UID,
4      /// Hit points before the boar is slain
5      hp: u64,
6      /// Strength of this particular boar
7      strength: u64,
8      /// An ID of the game
9      game_id: ID,
10 }
```



```
1  /// An immutable object that contains information about the  
2  /// game admin. Created only once in the module initializer,  
3  /// hence it cannot be recreated or falsified.  
4  struct GameInfo has key {  
5      id: UID,  
6      admin: address  
7  }
```



```
1  /// Capability conveying the authority to create boars and potions
2  struct GameAdmin has key {
3      id: UID,
4      /// Total number of boars the admin has created
5      boars_created: u64,
6      /// Total number of potions the admin has created
7      potions_created: u64,
8      /// ID of the game where current user is an admin
9      game_id: ID,
10 }
```

Events

Wait... what are events?





```
1  /// Event emitted each time a Hero slays a Boar
2  struct BoarSlainEvent has copy, drop {
3      /// Address of the user that slayed the boar
4      slayer_address: address,
5      /// ID of the Hero that slayed the boar
6      hero: ID,
7      /// ID of the now-deceased boar
8      boar: ID,
9      /// ID of the game where event happened
10     game_id: ID,
11 }
```



```
1  /// Upper bound on player's HP  
2  const MAX_HP: u64 = 1000;  
3  /// Upper bound on how magical a sword can be  
4  const MAX_MAGIC: u64 = 10;  
5  /// Minimum amount you can pay for a sword  
6  const MIN_SWORD_COST: u64 = 100;
```




```
1  // TODO: proper error codes
2  /// The boar won the battle
3  const EBOAR_WON: u64 = 0;
4  /// The hero is too tired to fight
5  const EHERO_TIRED: u64 = 1;
6  /// Trying to initialize from a non-admin account
7  const ENOT_ADMIN: u64 = 2;
8  /// Not enough money to purchase the given item
9  const EINSUFFICIENT_FUNDS: u64 = 3;
10 /// Trying to remove a sword, but the hero does not have one
11 const ENO_SWORD: u64 = 4;
12 /// Assertion errors for testing
13 const ASSERT_ERR: u64 = 5;
```

Initialization



```
1  /// On module publish, sender creates a new game. But once it is published,  
2  /// anyone create a new game with a `new_game` function.  
3  fun init(ctx: &mut TxContext) {  
4      create(ctx);  
5  }
```



```
1  /// Anyone can create run their own game, all game objects will be  
2  /// linked to this game.  
3  public entry fun new_game(ctx: &mut TxContext) {  
4      create(ctx);  
5  }
```



```
1  /// Create a new game. Separated to bypass public entry vs init requirements.
2  fun create(ctx: &mut TxContext) {
3      let sender = tx_context::sender(ctx);
4      let id = object::new(ctx);
5      let game_id = object::uid_to_inner(&id);
6
7      transfer::freeze_object(GameInfo {
8          id,
9          admin: sender,
10     });
11
12     transfer::transfer(
13         GameAdmin {
14             game_id,
15             id: object::new(ctx),
16             boars_created: 0,
17             potions_created: 0,
18         },
19         sender
20     )
21 }
```

Gameplay

```

1  /// Slay the 'boar' with the 'hero's sword, get experience.
2  /// Aborts if the hero has 0 HP or is not strong enough to slay the boar
3  public entry fun slay(
4      game: &GameInfo, hero: &mut Hero, boar: Boar, ctx: &TxContext
5  ) {
6      check_id(game, hero.game_id);
7      check_id(game, boar.game_id);
8      let Boar { id: boar_id, strength: boar_strength, hp, game_id: _ } = boar;
9      let hero_strength = hero_strength(hero);
10     let boar_hp = hp;
11     let hero_hp = hero.hp;
12     // attack the boar with the sword until its HP goes to zero
13     while (boar_hp > hero_strength) {
14         // first, the hero attacks
15         boar_hp = boar_hp - hero_strength;
16         // then, the boar gets a turn to attack. if the boar would kill
17         // the hero, abort—we can't let the boar win!
18         assert!(hero_hp >= boar_strength , EBOAR_WON);
19         hero_hp = hero_hp - boar_strength;
20
21     };
22     // hero takes their licks
23     hero.hp = hero_hp;
24     // hero gains experience proportional to the boar, sword grows in
25     // strength by one (if hero is using a sword)
26     hero.experience = hero.experience + hp;
27     if (option::is_some(&hero.sword)) {
28         level_up_sword(option::borrow_mut(&mut hero.sword), 1)
29     };
30     // let the world know about the hero's triumph by emitting an event!
31     event::emit(BoarSlainEvent {
32         slayer_address: tx_context::sender(ctx),
33         hero: object::uid_to_inner(&hero.id),
34         boar: object::uid_to_inner(&boar_id),
35         game_id: id(game)
36     });
37     object::delete(boar_id);
38 }

```



```
1  check_id(game, hero.game_id);
2  check_id(game, boar.game_id);
3  let Boar { id: boar_id, strength: boar_strength, hp, game_id: _ } = boar;
4  let hero_strength = hero_strength(hero);
5  let boar_hp = hp;
6  let hero_hp = hero.hp;
7  // attack the boar with the sword until its HP goes to zero
8  while (boar_hp > hero_strength) {
9      // first, the hero attacks
10     boar_hp = boar_hp - hero_strength;
11     // then, the boar gets a turn to attack. if the boar would kill
12     // the hero, abort--we can't let the boar win!
13     assert!(hero_hp >= boar_strength , EBOAR_WON);
14     hero_hp = hero_hp - boar_strength;
15
16 };
17 // hero takes their licks
18 hero.hp = hero_hp;
```




```
1 // hero gains experience proportional to the boar, sword grows in
2 // strength by one (if hero is using a sword)
3 hero.experience = hero.experience + hp;
4 if (option::is_some(&hero.sword)) {
5     level_up_sword(option::borrow_mut(&mut hero.sword), 1)
6 };
7 // let the world know about the hero's triumph by emitting an event!
8 event::emit(BoarSlainEvent {
9     slayer_address: tx_context::sender(ctx),
10     hero: object::uid_to_inner(&hero.id),
11     boar: object::uid_to_inner(&boar_id),
12     game_id: id(game)
13 });
14 object::delete(boar_id);
```




```
1  /// Strength of the hero when attacking
2  public fun hero_strength(hero: &Hero): u64 {
3      // a hero with zero HP is too tired to fight
4      if (hero.hp == 0) {
5          return 0
6      };
7
8      let sword_strength = if (option::is_some(&hero.sword)) {
9          sword_strength(option::borrow(&hero.sword))
10     } else {
11         // hero can fight without a sword, but will not be very strong
12         0
13     };
14     // hero is weaker if he has lower HP
15     (hero.experience * hero.hp) + sword_strength
16 }
17
18 fun level_up_sword(sword: &mut Sword, amount: u64) {
19     sword.strength = sword.strength + amount
20 }
21
22 /// Strength of a sword when attacking
23 public fun sword_strength(sword: &Sword): u64 {
24     sword.magic + sword.strength
25 }
```

Inventory



```
1  /// Heal the weary hero with a potion
2  public fun heal(hero: &mut Hero, potion: Potion) {
3      assert!(hero.game_id == potion.game_id, 403);
4      let Potion { id, potency, game_id: _ } = potion;
5      object::delete(id);
6      let new_hp = hero.hp + potency;
7      // cap hero's HP at MAX_HP to avoid int overflows
8      hero.hp = math::min(new_hp, MAX_HP)
9  }
```



```
1  /// Add `new_sword` to the hero's inventory and return the old sword  
2  /// (if any)  
3  public fun equip_sword(hero: &mut Hero, new_sword: Sword): Option<Sword> {  
4      option::swap_or_fill(&mut hero.sword, new_sword)  
5  }
```



```
1  /// Disarm the hero by returning their sword.
2  /// Aborts if the hero does not have a sword.
3  public fun remove_sword(hero: &mut Hero): Sword {
4      assert!(option::is_some(&hero.sword), ENO_SWORD);
5      option::extract(&mut hero.sword)
6  }
```

Object Creation



```
1  /// It all starts with the sword. Anyone can buy a sword, and proceeds go
2  /// to the admin. Amount of magic in the sword depends on how much you pay
3  /// for it.
4  public fun create_sword(
5      game: &GameInfo,
6      payment: Coin<SUI>,
7      ctx: &mut TxContext
8  ): Sword {
9      let value = coin::value(&payment);
10     // ensure the user pays enough for the sword
11     assert!(value >= MIN_SWORD_COST, EINSUFFICIENT_FUNDS);
12     // pay the admin for this sword
13     transfer::transfer(payment, game.admin);
14
15     // magic of the sword is proportional to the amount you paid, up to
16     // a max. one can only imbue a sword with so much magic
17     let magic = (value - MIN_SWORD_COST) / MIN_SWORD_COST;
18     Sword {
19         id: object::new(ctx),
20         magic: math::min(magic, MAX_MAGIC),
21         strength: 1,
22         game_id: id(game)
23     }
24 }
```




```
1 public entry fun acquire_hero(  
2     game: &GameInfo, payment: Coin<SUI>, ctx: &mut TxContext  
3 ) {  
4     let sword = create_sword(game, payment, ctx);  
5     let hero = create_hero(game, sword, ctx);  
6     transfer::transfer(hero, tx_context::sender(ctx))  
7 }
```




```
1  /// Anyone can create a hero if they have a sword. All heroes start with the  
2  /// same attributes.  
3  public fun create_hero(  
4      game: &GameInfo, sword: Sword, ctx: &mut TxContext  
5  ): Hero {  
6      check_id(game, sword.game_id);  
7      Hero {  
8          id: object::new(ctx),  
9          hp: 100,  
10         experience: 0,  
11         sword: option::some(sword),  
12         game_id: id(game)  
13     }  
14 }
```



```
1  /// Admin can create a potion with the given `potency` for `recipient`
2  public entry fun send_potion(
3      game: &GameInfo,
4      potency: u64,
5      player: address,
6      admin: &mut GameAdmin,
7      ctx: &mut TxContext
8  ) {
9      check_id(game, admin.game_id);
10     admin.potions_created = admin.potions_created + 1;
11     // send potion to the designated player
12     transfer::transfer(
13         Potion { id: object::new(ctx), potency, game_id: id(game) },
14         player
15     )
16 }
```



```
1  /// Admin can create a boar with the given attributes for `recipient`
2  public entry fun send_boar(
3      game: &GameInfo,
4      admin: &mut GameAdmin,
5      hp: u64,
6      strength: u64,
7      player: address,
8      ctx: &mut TxContext
9  ) {
10     check_id(game, admin.game_id);
11     admin.boars_created = admin.boars_created + 1;
12     // send boars to the designated player
13     transfer::transfer(
14         Boar { id: object::new(ctx), hp, strength, game_id: id(game) },
15         player
16     )
17 }
```

Game Integrity / Links Checks



```
1 public fun check_id(game_info: &GameInfo, id: ID) {  
2     assert!(id(game_info) == id, 403); // TODO: error code  
3 }  
4  
5 public fun id(game_info: &GameInfo): ID {  
6     object::id(game_info)  
7 }
```

Bibliography/ Further Reading

[github.com/MystenLabs/sui/blob/main/sui_programmability/examples/games/
sources/hero.move](https://github.com/MystenLabs/sui/blob/main/sui_programmability/examples/games/sources/hero.move)

What's Next!

Next Workshop: Project Showcase: **On-chain RPG** (Part 2)

March 24, 2023

Sui Vietnam Builder House!

lu.ma/sui.vietnam

Survey + Questions?

Twitter: @0xShayan



